SPEDAS training session
- advanced course -

T. Hori, S. Kurita, C.-W. Jun, M. Shoji, and Y. Miyoshi
(ERG Science Center, ISEE, Nagoya Univ.)

Goal of this training

- We are going to learn how to load, plot, and manipulate LEP-e data of the ERG satellite, with a special focus on analysis of observed electron distribution functions.

- For this purpose, we try to reproduce various plots published in Kazama et al., GRL, 2018.

Contents of today's course

- Read/plot Lv.3 prov. data
- Quick review of how to use part_products with LEP-e data
- HFA specs and PA-limited E-t specs of LEP-e (like Figure 1 of Kazama et al. [2018])
- PA-energy spectra and their line plot (like Figures 2 and 3)
- Density and temperature by part_products and Maxwellian-fitting (like Figure 4)

Keep in mind upon the training

- This is a "hands-on" training for SPEDAS, not a time for e-mail check!
- Communicate with lecturers, tutors, and neighboring skilled users.
- Today's session might not be able to cover all topics in the handouts due to time limitation. It is recommended to go through the entire contents later by yourself.
- We will proceed rather slowly with intermediate-level users, but you can practice at your own pace.

Keep in mind upon the training (cont'd)

- The band width of the available wireless networks is quite limited today. For a smooth and successful training, please be sure to follow the rules listed below, when you are hooked up to the WiFi networks for the training:
  - Stop using/synchronizing online storages (Dropbox, Google drive, OneDrive, iCloud, etc.) to **save the band width of the network**.
  - Turn off Windows Update temporarily (Windows users).
  - Do not access online movie sites (Youtube, Niko-Niko-Douga(ニコニコ動画), etc.).

Before starting: Preparation of data files

- We have asked you to download necessary data files in advance, BEFORE you come here.

- The data files should be saved under the data root directory returned by the function `root_data_dir()`.
  - Type `print, root_data_dir()` on IDL then your data root directory will be shown.

Before starting: Choice of the color table

- In this handouts, all figures are drawn with the color table "JET".
- But, of course, you can do this training with your own favorite color scheme.

JET




```
IDL> loadct_sd, 46
```

FAST-special (default)




```
loadct2, 43
```

viridis




```
loadct_sd, 47
```

`loadct2` accepcts table numbers of 0–74.

Brief introduction on a newly arrived data set: LEP-e Level-3 provisinal data

Provisional Lv.3 data of LEP-e

```
ERG> timespan,  '2017-05-01'
ERG> erg_load_lepe_pa, uname=uname, pass=pass
ERG> tplot, 'erg_lepe_l3_prov_PA_' + ['energy_15', 'pitchangle_09' ]
```

Please type the command lines without the command prompt "ERG>". The prompt is omitted from the next slide on so that one can just copy&paste commands on an IDL console.



Quick review of the usages of the part_products library

What's "part_products"?

- A set of generic routines bundled to SPEDAS to make tplot variables for various types of spectrum plot.



Basic usage of part_products: Energy spectra of average flux

```
timespan, '2017-04-12'
erg_load_lepe, datatype='3dflux', uname=uname, pass=pass, /no_sort_enebin

erg_lep_part_products, 'erg_lepe_l2_3dflux_FEDU', output='energy'
tplot, 'erg_lepe_l2_3dflux_FEDU_energy'
```



Basic usage of part_products: Pitch angle spectra

```
timespan, '2017-04-12/04:00',8, /hour & get_timespan, tr
erg_load_mgf & set_erg_var_label
magvn = 'erg_mgf_l2_mag_8sec_dsi' & posvn = 'erg_orb_l2_pos_gse'

erg_lep_part_products, 'erg_lepe_l2_3dflux_FEDU', output='pa', $
       energy=[1000., 1500.], trange=tr, mag=magvn, pos=posvn
tplot, 'erg_lepe_l2_3dflux_FEDU_*'
```



Basic usage of part_products: Energy spectra for a limited PA range

```
erg_lep_part_products, 'erg_lepe_l2_3dflux_FEDU', output='energy', $
       trange=tr, mag=magvn, pos=posvn, pitch=[80, 100]
tplot, 'erg_lepe_l2_3dflux_FEDU_*'
```



Basic usage of part_products: Velocity moments

```
erg_lep_part_products, 'erg_lepe_l2_3dflux_FEDU', $
        output='moments', mag=magvn,pos=posvn, trange=tr

tplot, 'erg_lepe_l2_3dflux_FEDU_' + $
                    ['energy','pa','ptens', 'density']
```



Reproduction of the figures of Kazama et al. [GRL, 2018]

Figure 1 of Kazama+2018

```
timespan, '2017-04-12/16:00', 2, /hour
get_timespan, tr
erg_load_lepe, datatype='3dflux', uname=uname, pass=pass, /no_sort_enebin
erg_load_mgf
set_erg_var_label
erg_lep_part_products, 'erg_lepe_l2_3dflux_FEDU', output='energy'
erg_load_pwe_hfa, datatype='spec'
erg_load_pwe_ofa, datatype='spec'


options, 'erg_pwe_hfa_l2_lm_spectra_e_mix', zrange=[1e-8, 1e-2], zlog=1, ytitle='PWE/HFA!C(E_MIX)', $
     ysubtitle='[Hz]', auto_downsample=1
options, 'erg_pwe_ofa_spec_l2_E_spectra_merged', zrange=[1e-8, 1e-2], zlog=1, ytitle='PWE/OFA!Cspec', $
     ysubtitel='[Hz]', auto_downsample=1


tplot, ['erg_pwe_hfa_l2_lm_spectra_e_mix', 'erg_pwe_ofa_spec_l2_E_spectra_merged', 'erg_lepe_l2_3dflux_FEDU_energy' ]
```



Figure 1 of Kazama+2018 (cont'd)

```
erg_lep_part_products, 'erg_lepe_l2_3dflux_FEDU', output='energy', $
       trange=tr, mag=magvn, pos=posvn, pitch=[0, 15], suffix='_pa00-15'
erg_lep_part_products, 'erg_lepe_l2_3dflux_FEDU', output='energy', $
       trange=tr, mag=magvn, pos=posvn, pitch=[85, 95], suffix='_pa85-95'
zlim, ['erg_lepe_l2_3dflux_FEDU_energy_pa*'], 1e+2, 1e+6, 1


tplot, [ 'erg_pwe_hfa_l2_lm_spectra_e_mix', $
'erg_pwe_ofa_spec_l2_E_spectra_merged', $
 'erg_lepe_l2_3dflux_FEDU_energy_pa*'  ]
```



Figure 2 of Kazama+2018

```
dists = erg_lepe_get_dist( 'erg_lepe_l2_3dflux_FEDU', trange=tr )


erg_part_en_pa_spec_plot, dists[450], units='flux'


erg_part_en_pa_spec_plot, dists, time='2017-04-12/17:00:00', units='eflux'


erg_part_en_pa_spec_plot, dists, $
      time=['2017-04-12/17:05:35','2017-04-12/17:06:08'], $
      /with_contour, units='df_km'
```



Figure 2 of Kazama+2018 (cont'd)

```
erg_part_en_pa_spec_plot, dists, $
       time=['2017-04-12/17:05:35','2017-04-12/17:06:08'], $
       /noplot, rslt=rslt, units='df_km'
help, rslt


yid = nn2( rslt.y_pitchangle, 30. ) ;; for PA=30 deg
plot, rslt.x_energy, rslt.z_hist[*, yid], /xlog, /ylog, psym=-6
yid2 = nn2( rslt.y_pitchangle, 90 ) ;; for PA=90 deg
oplot, rslt.x_energy, rslt.z_hist[*, yid2], psym=-6,
color=spd_get_color('purple')
```

```
ERG> help, rslt
** Structure <2191318>, 3 tags, length=2476,
   X_ENERGY        FLOAT     Array[30]
   Y_PITCHANGLE    FLOAT     Array[19]
   Z_HIST          FLOAT     Array[30, 19]
ERG>
```



Figure 2 of Kazama+2018 (cont'd)

```
erg_part_maxwellian_fit, rslt.x_energy[2:14], rslt.z_hist[2:14, yid2], n90, kbt90
df_km = erg_part_get_maxwellian( rslt.x_energy, n90, kbt90 )
oplot, rslt.x_energy, df_km, color=spd_get_color('purple'), $
    linestyle=2, thick=2.5
print, 'n = ', n90, ' [/cc]', '   kBT = ', kbt90,  ' [eV]'

erg_part_maxwellian_fit, rslt.x_energy[2:14], rslt.z_hist[2:14, yid], n30, kbt30
df_km = erg_part_get_maxwellian( rslt.x_energy, n30, kbt30 )
oplot, rslt.x_energy, df_km, color=spd_get_color('black'), $
    linestyle=2, thick=2.5
print, 'n = ', n30, ' [/cc]', '   kBT = ', kbt30,  ' [eV]'
```

```
ERG> print, 'n = ', n90, ' [/cc]', '   kBT = ', kbt90,  ' [eV]'
n =       0.46398556 [/cc]   kBT =        137.79072 [eV]
ERG>
ERG> print, 'n = ', n30, ' [/cc]', '   kBT = ', kbt30,  ' [eV]'
n =       0.11406248 [/cc]   kBT =        203.69873 [eV]
ERG>
```

```
ERG> print, rslt.x_energy[ [2,14] ]
      67.0372       963.422
ERG>
```



Figure 3 of Kazama+2018

```
timespan, '2017-04-12/16:50', 40, /min & get_timespan, tr
erg_lep_part_products, 'erg_lepe_l2_3dflux_FEDU', output='pa', units='df_km', $
        energy=[90., 92.], suffix='_91ev', trange=tr, mag=magvn, pos=posvn
erg_lep_part_products, 'erg_lepe_l2_3dflux_FEDU', output='pa', units='df_km', $
        energy=[341., 343.], suffix='_342ev', trange=tr, mag=magvn, pos=posvn

erg_load_pwe_efd, datatype='pot'

tplot, [  'erg_pwe_efd_l2_Vave', 'erg_lepe_l2_3dflux_FEDU_pa_342ev', $
          'erg_lepe_l2_3dflux_FEDU_pa_91ev' ]
```



Figure 3 of Kazama+2018 (cont'd)

```
erg_lep_part_products, 'erg_lepe_l2_3dflux_FEDU', output='energy', units='eflux', $
                       pitch=[85., 95.], suffix='_pa85-95', trange=tr, mag=magvn, pos=posvn
erg_lep_part_products, 'erg_lepe_l2_3dflux_FEDU', output='energy', units='eflux', $
                       pitch=[0., 15.], suffix='_pa00-15', trange=tr, mag=magvn, pos=posvn

get_data,  'erg_lepe_l2_3dflux_FEDU_energy_pa85-95', data=dperp, dl=dl, lim=lim
get_data,  'erg_lepe_l2_3dflux_FEDU_energy_pa00-15', data=dpara, dl=dl, lim=lim

id_342ev = nn( dperp.v[0, *], 342. )
id_91ev = nn( dperp.v[0, *], 91 )
store_data, 'erg_lepe_eflux_342ev', data={x:dperp.x, y:[ [dperp.y[*, id_342ev]], [dpara.y[*, id_342ev]] ] }
store_data, 'erg_lepe_eflux_91ev', data={x:dperp.x, y:[ [dperp.y[*, id_91ev]], [dpara.y[*, id_91ev]] ] }
ylim, 'erg_lepe_eflux_*ev', 0, 0, 0
options, 'erg_lepe_eflux_*ev', labels=['perp', 'para'], colors=[2, 187], ysubtitle=dl.ztitle

tplot, ['erg_pwe_efd_l2_Vave', 'erg_lepe_eflux_342ev', 'erg_lepe_l2_3dflux_FEDU_pa_342ev', 'erg_lepe_eflux_91ev', 'erg_lepe_l2_3dflux_FEDU_pa_91ev' ]
```



Figure 4 of Kazama+2018

```
erg_load_pwe_hfa, level='l3'
ylim, 'erg_pwe_hfa_l2_lm_spectra_e_mix', 10., 100., 1
zlim, 'erg_pwe_hfa_l2_lm_spectra_e_mix', 1e-9, 1e-2, 1

calc, '"erg_pwe_efd_ne_vsc" = 26.39 * exp( -0.6546 * "erg_pwe_efd_l2_Vave" )'

erg_lep_part_products, 'erg_lepe_l2_3dflux_FEDU', output='moments', energy=[40., 750.], mag=magvn, pos=posvn, trange=tr
ylim, 'erg_lepe_l2_3dflux_FEDU_magt3', 1e+1, 1e+3, 1

store_data, 'erg_lepe_efd_density', data=['erg_pwe_efd_ne_vsc', 'erg_lepe_l2_3dflux_FEDU_density' ]
ylim, 'erg_lepe_efd_density', 1e-2, 1e+2, 1
options, 'erg_lepe_efd_density', ytitle='Density', ysubtitle='[/cc]', colors=[2, 6], $
      labels=['total', 'hot'], labflag=-1, constant=[0.1, 1., 10.]

tplot, [ 'erg_pwe_hfa_l3_ne_mgf', 'erg_pwe_hfa_l2_lm_spectra_e_mix', 'erg_lepe_efd_density', 'erg_lepe_l2_3dflux_FEDU_magt3' ]
```



Comparison of fitted parameters and integrated parameters

[Kazama+2018]



The hot density and temperatures were derived with the bi-Maxwellian fitting.

by part_products



The same parameters have been derived by just integrating 3-D distribution functions.

Exercise

If you have completed too early, please fill the remaining time by trying the following exercise.

- Do Maxwellian fitting for LEP-e data for the entire time interval of 16:50–17:30 to derive and then plot time series of the fitted number density and perpendicular temperature.
  - Please fit to distribution functions in the energy range of 67–1000 eV.
  - You only have to calculate only perpendicular density and temperature.
  - You could combine the following commands:
    - `erg_lep_part_products, get_data, erg_part_maxwellian_fit, store_data, tplot, …`



Appendix

Appendix A-1:
List of velocity moments by part_products

(copied from the handouts of the training session of last year)

```
del_data, '*'
timespan, '2017-03-27/10:00', 1, /hour  & get_timespan, tr
erg_load_mepe, datatype='3dflux', varformat='FEDU'
erg_load_mepi_nml, datatype='3dflux', varformat='FPDU'
erg_load_mgf & erg_load_orb
erg_mep_part_products, 'erg_mepi_l2_3dflux_FPDU', pos='erg_orb_l2_pos_gse',
mag='erg_mgf_l2_mag_8sec_dsi', output='moments', trange=tr
erg_mep_part_products, 'erg_mepe_l2_3dflux_FEDU', pos='erg_orb_l2_pos_gse',
mag='erg_mgf_l2_mag_8sec_dsi', output='moments', trange=tr
```

```
ERG> tplot_names, 'erg_mepi_l2_3dflux_FPDU_*'
  47 erg_mepi_l2_3dflux_FPDU_avgtemp
  48 erg_mepi_l2_3dflux_FPDU_density
  49 erg_mepi_l2_3dflux_FPDU_eflux
  50 erg_mepi_l2_3dflux_FPDU_flux
  51 erg_mepi_l2_3dflux_FPDU_mftens
  52 erg_mepi_l2_3dflux_FPDU_ptens
  53 erg_mepi_l2_3dflux_FPDU_sc_current
  54 erg_mepi_l2_3dflux_FPDU_velocity
  55 erg_mepi_l2_3dflux_FPDU_vthermal
  56 erg_mepi_l2_3dflux_FPDU_magf
  57 erg_mepi_l2_3dflux_FPDU_magt3
  58 erg_mepi_l2_3dflux_FPDU_t3
  59 erg_mepi_l2_3dflux_FPDU_sc_pot
  60 erg_mepi_l2_3dflux_FPDU_symm
  61 erg_mepi_l2_3dflux_FPDU_symm_theta
  62 erg_mepi_l2_3dflux_FPDU_symm_phi
  63 erg_mepi_l2_3dflux_FPDU_symm_ang
ERG>
```

Primary parameters calculated with the part_products:

- density: number density
- avgtemp: scalar temperature (!)
- velocity: bulk velocity
- vthermal: thermal velocity
- mtens: momentum flux density tensor
- ptens: pressure tensor
- t3: temperature tensor (!)
- magt3: perpendicular/parallel temperature (!)
- flux: number flux
- eflux: energy flux

All vector and tensor quantities in DSI coordinates.

(!) Note that these are NOT a temperature defined as a width of Maxwellian distribution.

Appendix A-2:
3-D data structure common to particle data that SPEDAS can handle

```
ERG> help, dists[0]
** Structure <18a6808>, 21 tags, length=196736, data length=196725, refs=2:
   PROJECT_NAME    STRING    'ERG'
   SPACECRAFT      LONG                 1
   DATA_NAME       STRING    'LEP-e Electron 3dflux'
   UNITS_NAME      STRING    'flux'
   UNITS_PROCEDURE STRING    'erg_convert_flux_units'
   SPECIES         STRING    'e'
   VALID           BYTE         1
   CHARGE          FLOAT          -1.00000
   MASS            FLOAT       5.68566e-06
   TIME            DOUBLE       1.4920128e+09
   END_TIME        DOUBLE       1.4920128e+09
   DATA            FLOAT     Array[32, 16, 12]
   BINS            FLOAT     Array[32, 16, 12]
   ENERGY          FLOAT     Array[32, 16, 12]
   DENERGY         FLOAT     Array[32, 16, 12]
   NENERGY         LONG                 32
   NBINS           LONG                192
   PHI             FLOAT     Array[32, 16, 12]
   DPHI            FLOAT     Array[32, 16, 12]
   THETA           FLOAT     Array[32, 16, 12]
   DTHETA          FLOAT     Array[32, 16, 12]
ERG>
```

An example for LEP-e 3-D flux data:

`dist` is an array of structures each of which contains a set of data for each spin.

"DATA" holds the flux data as a 3-D array of 32 ene. ch x 16 spin sector x 12 sensors for this case.

ENERGY and DENERGY are the central energies and energy ranges of the energy channels.

PHI, DPHI, THETA, and DTHETA have phi/theta angles of particle-going directions and angular widths measured by directional channels of a particle instrument in the DSI coordinate system.

Appendix A-3:
Some other options available for erg_part_en_pa_spec_plot

```
erg_part_en_pa_spec_plot, dist $
  , time=time $       ; a time or time range for plotting
  , units=units $     ; physical unit 'flux','eflux','df_km','df_cm'
  , with_contour=with_contour $  ; to overlay contour lines
  , zrange=zrange $   ; explicitly set the range for the color scale
  , npabin=npabin $   ; number of pitch angle bins (default: 19)
  , rslt=rslt $       ; to obtain data arrays which have been plotted
  , noplot=noplot     ; set to suppress replotting
```

You can use this for other particle data.
For example:

```
timespan, '2017-04-12/16:00', 2, /hour
get_timespan, tr
erg_load_mepe, datatype='3dflux'
dists = erg_mepe_get_dist( $
          'erg_mepe_l2_3dflux_FEDU', trange=tr)
erg_part_en_pa_spec_plot, dists
```



Appendix A-4:
ISEE3D: a 3-D distribution function viewer





Now ISEE3D supports LEP-e data!

Detailed functions of the tool were described by Keika et al. [EPS, 2018].

```
ERG> erg_crib_lepe_isee3d, $
       trange='2017-04-12/'+['16:00','17:30']
```